ISSN 0193-3728

A Basic Model for Use in Computer Simulations of *Boophilus* Tick Biology and Control



U.S. Department of Agriculture
Agricultural Research Service
Advances in Agricultural Technology • AAT-S-32/December 1983

This publication is available from the Insects Affecting Man and Animals Research Laboratory, P.O. Box 14565, Gainesville, Fla. 32604.

Agricultural Research Service, Advances in Agricultural Technology, Southern Series, No. 32, December 1983.

Published by Agricultural Research Service (Southern Region), U.S. Department of Agriculture, P.O. Box 53326, New Orleans, La. 70153.

CONTENTS

ILLUSTRATIONS

TABLES

A Basic Model for Use in Computer Simulations of *Boophilus* Tick Biology and Control

By D. E. Weidhaas,[1] D. G. Haile,[2] J. E. George,[3] R. L. Osburn,[4] and R. O. Drummond[5]

ABSTRACT

A life-history model of *Boophilus* ticks was adapted to computer-simulation techniques to reproduce the densities of tick populations as they are observed to occur under natural conditions. The simulation model was then used to study the interaction of density and control methods. Index terms: biological control (ticks), *Boophilus annulatus* (Canestrini), *Boophilus microplus* (Say), *Boophilus* spp., cattle tick, chemical control (ticks), computer models, life-history models, livestock ticks, models, pest control, southern cattle tick, ticks.

INTRODUCTION

The cattle tick, *Boophilus annulatus* (Canestrini), and the southern cattle tick, *Boophilus microplus* (Say), were major deterrents to cattle production in the Southern United States before their eradication during the first half of the 1900's (Graham and Hourrigan 1977). These ticks are still a threat to U.S. livestock producers through possible reinvasion across the United States-Mexican border. A quarantine that includes livestock inspection and treatment is maintained along this border to prevent such reinvasion. Furthermore, eradication campaigns against *Boophilus* ticks are underway in Mexico and Puerto Rico. So there is a need to better understand the population dynamics of these ticks as they occur in these areas and how the populations respond to the control methods applied to eradicate them. How the control methods affect population density over a long time is sufficiently complex to require not only extensive biological and ecological field research but also the assistance of computer methods to quantify the interactions.

Descriptive models of the life histories of *Boophilus* ticks are available. Sutherst et al. (1978) present excellent diagrammatic schemes of the life histories of representative types of ticks, including one-host ticks. Sutherst and Wharton (1973) and Sutherst and Dallwitz (1979) discuss development of a model for *B. microplus* adapted to computer simulations. In their model, they quantify the life history by dealing with three major components: larval productivity by engorged females in pastures, the longevity and host-finding rate of larvae that are off the host, and the mortality of parasite stages on cattle. They use meteorological data to drive the model and mathematical expressions of key biological interactions to control it. Their stated objective is to answer "three basic questions, namely: what factors [are] responsible for fluctuations, regula-

[1]Laboratory director, Insects Affecting Man and Animals Research Laboratory, Agricultural Research Service, U.S. Department of Agriculture, P.O. Box 14565, Gainesville, Fla. 32604.

[2]Research engineer, Insects Affecting Man and Animals Research Laboratory.

[3]Research leader, Tick Research Unit, U.S. Livestock Insects Laboratory, Agricultural Research Service, U.S. Department of Agriculture, P.O. Box 232, Kerrville, Tex. 78028.

[4]Head, Department of Biology, Georgia Southern University, Statesboro, Ga. 30460.

[5]Laboratory director, U.S. Livestock Insects Laboratory.

tions and maintenance of equilibrium levels of tick populations." They have produced an excellent model and have identified major biological and meteorological interactions and the importance of cattle breed, immunity, and density-dependent regulation on the development of the parasitic stages.

Osburn and Knipling (1982) describe a mathematical model that analyzes the potential of the release of sterile hybrid males and fertile hybrid females from crosses of *B. annulatus* and *B. microplus* (Thompson et al. 1981) for eradication. Their model highlights the dynamics of uncontrolled low-level populations as well as populations subjected to the release of hybrids. They use only population growth rate and survival of various stages over periods of time approximating the generation times. By describing and quantifying the interaction of released hybrids and native ticks, they have been able to calculate predicted decreases in density of native ticks that would be caused by hybrid releases. Their model is valid since they have identified the major factors controlling tick density, but it is laborious because it requires repetitive numerical calculations. Also, their model does not allow easy analysis of the effect of insecticide treatments.

We identified a need for computer-simulation techniques based on life-history analysis of *Boophilus* ticks that would permit us to theorize and analyze the effectiveness of additional control methods for eradication schemes for these ticks. Our needs were not to study factors that regulate populations at equilibrium as Sutherst emphasized but to determine the effect of patterns and variations in control treatments applied to populations below equilibrium as they proceed to eradication. This paper describes a dynamic life-history model for *Boophilus* species that combines some features of the models of both approaches cited above. It follows the general approach of Haile and Weidhaas (1977) for relating population density to control treatments through computer simulations based on life-history parameters. It also provides speed and ease of computer analysis to the control model described by Osburn and Knipling (1982) with less biological and meteorological data than used by Sutherst and colleagues. In addition to discussing development and initialization of the model, we have included the FORTRAN-WATFIV program for the most basic simulation used to verify the accuracy of the model. The model as it is adapted to computer simulation through the FORTRAN program allows comparison of the effects of insecticidal control methods and of sterile-hybrid release under simulated situations. In its present form, it provides a basis for further model development when additional variables and data are added to it.

MODEL DEVELOPMENT

Our system is designed as a life-history model that describes all the developmental stages of the tick—eggs, larvae, nymphs, and adults—subdivided into weekly age classes when they are either on or off the host. As will be evident, we have deliberately finessed four major considerations in the construction of a model describing the dynamics of tick populations: (1) how temperature and humidity affect the hatching of eggs and the survival, developmental, and host-seeking rates of the stages off the host; (2) any temperature-activated hibernation or estivation; (3) density-dependent regulation of development of the parasitic stages; and (4) the resistance of cattle or breeds of cattle to ticks.

We designed the model in this way for several reasons: Awaiting the development of such data from the many necessary locations would preclude not only development of the model but also its usefulness in identifying and solving problems. Routines to make the model more complex can be added when the data become available. Some of the effects of these factors can be handled adequately, at least for our purpose, by averages of key parameters. When our prime interest is the control of tick populations, particularly at low density levels leading to eradication, density-dependent regulation can be considered to be at its minimum value. Identification of key controlling parameters can be used to include the effects of as yet unquantified variables. And finally, if a high level of control or eradication is the major objective of the analysis, it is much more important to know the density of all the ticks than to know their distribution among their animal hosts.

Figure 1, under the heading "Wild Ticks," is a model of their life history. It provides for differentiating eggs, larvae, and adult females off the host and larvae, nymphs, and adult males and females on the host. The structure of the model is determined by the average development times of the various stages (in weeks) and the maximum



FIGURE 1.—Life-history model for wild and hybrid ticks. Each box represents 1 week.

survival time of larvae off the host under normal conditions in weeks. (A week was chosen as the interval since all actual development times approximated this time or multiples of it.) In this model, the times are held constant, but they could be made variable with temperature and humidity in further model development. The development times of the parasitic stages were taken from Hitchcock (1955), including his own data and that of others he summarized. The development times of the stages off the host (including maximum survival time of larvae under summertime temperatures) were based on the experience of tick specialists at Kerrville, Tex. The model provides 2 weeks for egg hatch; a maximum survival of 12 weeks for larvae off the host; 1 week each for development of larvae, nymphs, and adults on the host; and 3 weeks for egg deposition by females. In the model, ticks proceed through their development at weekly intervals. To make the life cycle complete, one needs only to decide on the proportion of larval ticks off the host that are picked up by the host and the average number of eggs laid per female. So we have a descriptive model of the dynamics of all stages of *Boophilus* tick populations. We need only identify key parameters and their values; this will allow us to supply numbers to the age classes of the various stages for initialization on the week a simulation is to begin and for updating the model weekly for the period of time it is to be run. The following key parameters for quantification were used:

Population growth rate per generation	$Ro=2.0$
Generation time in weeks	$G=7.77$
Weekly growth rate	$\lambda=1.0933$
Sex ratio	$Sr=1:1$
Number of eggs per female	$M=3{,}000$
Fraction of larvae picked up weekly by hosts.	$P=0.0114736$

Survival and mortality values were:

Stage	Survival			Mortality (%)
	Symbol	Average weekly	Total	
Eggs	*SE*	0.8660	0.75	25
Larvae off host.	*SL*	.495742	.00	100
Larvae, nymphs, adults on host.	*SH*	.6299605	.25	75
Engorged females off host.	*SF*	.5625	0	100

The sex ratio and the number of eggs per female are taken from laboratory studies on tick biology. The generation time is derived from our computer runs. All other parameters are arbitrary values based on judgment or calculations. The generation growth rate is set at a constant value of 2, the value used by Osburn and Knipling (1982). For our purposes and simulations, we will deal only with low-level tick populations much below the carrying capacity of host animals, and the growth rate of 2 is considered an average maximum. Therefore, no provision for density-dependent regulation is required. The weekly growth rate can be calculated by

$$\ln\lambda = (\ln Ro)/G.$$

Our values for the fraction of larvae picked up weekly by hosts and the survivals of the various stages on and off the host are arbitrary estimates of what they may actually be. Until such data are available for specified areas and conditions, such estimates are the only approach possible, and model realism will depend on model verification. But the values of all these parameters are controlled by the generation growth rate of the population. For more sophisticated models, the growth rate per generation or per week and survival values could be programed to change, allowing for variable density curves observed in specfic areas or locations.

In this model, we have allowed the males to survive only 1 week on the host. We realize that males can remain on the host longer than females and remate. But, this behavior is not critical to the purposes of the model. The effectiveness of insecticidal treatments is generally based on the number of females on the host. For simulations of the release of sterile hybrids, provision would have to be made for longer male survival if there were a difference between survival rates of hybrids and natives.

INITIALIZATION OF THE MODEL

Our first problem is to initialize the model on the week a simulation is to begin with numbers representing the stages and age classes that are present at that time. There are two options: start with only one tick stage present, for example, eggs off the host or adults (male and female) on

the host, or start with all stages and age classes present as would occur with overlapping generations. We have used the second option as follows.

Since we desire to equate this simulation as best as possible to field populations, we start by identifying the number of adult female ticks on the host. For example, assume that there are 1,000 animal hosts and that the average number of adult female ticks per host is 0.5. Then we have 500 adult females and 500 males; this provides a starting point. In the initialization process, the number of individuals in any age class is related to the number in any other age class by both survival and the growth rate; the number in any age class is obtained by multiplying the previous age class by the appropriate survival divided by λ. Appendix B gives actual numbers obtained in this initialization process. The 500 females are multiplied by SH/λ and then twice by SF/λ. The number of egg-laying females is multiplied by 3,000 to obtain the number of eggs. The process is continued to complete the cycle; larvae off the host must be summed and multiplied by the pickup rate to determine the number of larvae on the host. Accuracy can be checked by obtaining the starting number of females on the host.

Using mathematics developed for life-history analysis, we have calculated the starting numbers in all age classes from only the average number of females per host. Obviously, it would be ideal if we had all the life-table data on these ticks at every location. But such information is not and will not be available for a long time.

SIMULATION OF POPULATION DYNAMICS OVER TIME

Once the model structure has been initialized with numbers, the only problem remaining is to program a computer to move the tick population through time by updating the various age classes at weekly intervals. With a computer, such a task is relatively easy. The numbers in each age class are reduced by the appropriate survival value and advanced to replace the numbers in the next age classes (fig. 1). The number of eggs is obtained by multiplying the number of egg-laying females by the average number of eggs per female. The number of larvae on the ground must be summed and multiplied by the pickup rate to determine the number of larvae on the host. In this manner, the model is updated weekly and run for any predetermined length of time, and the desired output is either printed or stored. With proper computer programing and appropriate survival values, it is also possible to simulate densities of tick populations with constant or variable growth rates over time.

PROGRAMING CONTROL METHODS

The primary purpose of constructing this model was to permit the theoretical evaluation of control methods on tick density over time. So we added programing for simulating insecticide treatments and the release of sterile hybrids. Figure 1 can be used to illustrate the ease of programing a simulation involving insecticide treatments. Removal of adults (male and female), nymphs, and larvae from their storage locations proportional to the effectiveness of any insecticide treatment is all that is needed. For example, if a treatment kills 95% of these stages, then the numbers are reduced by 95%. Treatments can be programed to occur on various schedules or intervals. The overall effect on the various stages, including those not affected by the treatment, can be followed as both the affected and unaffected stages continue their development.

Providing for the release of sterile hybrids requires additional computer storage for the released ticks and any progeny of these released hybrids (fig. 1). To accomplish this, we added 12 storage boxes to the model for released hybrids to be present as unfed larvae on the ground. Ticks to be released are added to the first storage box and then survive through 12 weeks to be picked up by animal hosts. We set the rate of pickup of the released ticks to be double that of the wild ticks because Osburn and Knipling (1982) assume it would be. They argue that the selective distribution of hybrid larvae into native tick habitats would optimize the opportunities for host encounter. Rate of pickup can, of course, be set at any level; or the ticks could be released directly on animals. We have not yet provided for release directly on animals since it involves only the difference in the rate of pickup. For example, for every 100 ticks released, there would be 100 on the animals if released there or only 1 on the animals if released on the ground with a 1% pickup.

A complete life-history storage system similar

to that for wild ticks was provided for sterile hybrids. Released ticks picked up by animals follow through the standard life cycle, and the progeny can be accumulated and updated at weekly intervals. At each weekly interval, it is possible to calculate the ratio of sterile hybrid males to wild males and adjust the hatching of eggs and the resulting degree of sterility for both wild and hybrid types corresponding to that ratio. When the sterility value is stored for 3 weeks, the egg hatch can be adjusted for the effect of sterile mating at the proper point of the life cycle; females mated by sterile males on the host do not deposit these sterile eggs until 3 weeks later when they are on the ground.

Two problems still exist in this model. The first, the more serious, involves the distribution of ticks on animals, particularly when the number of ticks is smaller than the number of host animals. Since expectations are that the number of ticks per animal will be variable, with some animals having more than others, one would expect a simulation model to reflect this type of distribution. But the data to establish the variable distribution on animals are not available to us. So this model distributes ticks equally on all hosts when the number of ticks is greater than the number of animals. When the number of adult females (or males) is less than the number of animals, we assume that each infested host has one male and one female, so we can calculate the number of infested hosts. In this case, of course, the number of female ticks is equal to the number of infested hosts. This approach is similar to that of Osburn and Knipling (1982), but we did not use their concept of a cluster of ticks, and we used weekly intervals rather than their 3-month period. In the Osburn-Knipling model, when hybrids are released, there are always enough of them to distribute to all animals. So hybrids on animals with no wild ticks are wasted. The effect worsens as tick density decreases. It is interesting to note from the Osburn-Knipling model that this wasting of released and progeny hybrid ticks requires that releases be continued at the starting level throughout to maintain an approximately constant ratio close to the initial ratio.

The second problem was to decide a cutoff point for the model, that is, when ticks have been eliminated. In this model, a population is considered to have been eradicated when the number of wild larval ticks picked up by all host animals is less than 2—one male and one female.

The computer program used in this simulation model is reproduced as appendix A and includes comments throughout to identify code names and the functions of the various parts of the program.

SIMULATIONS RUN TO VERIFY THE MODEL

For our purposes, we decided to conduct three types of simulations to verify the accuracy of the model and to make judgments on its usefulness: (1) a tick population increasing at a constant growth rate, doubling each generation; (2) the same population subjected to the release of sterile hybrids at intervals of 1, 2, or 4 weeks; and (3) the same population exposed to insecticide treatments (animal spraying or dipping that would affect only the parasitic stages) at different levels of effectiveness and intervals of application. The first simulation would simply confirm the accuracy of the program since the number of ticks should increase at a constant rate, doubling each generation. The simulations with the release of sterile hybrids would add further confidence to the usefulness of the model if they agreed with the results of Osburn and Knipling (1982). Furthermore, we could examine how the interval of time between releases would affect the degree of control. Finally, simulations of insecticide treatments could be compared to results of their actual use over many years.

Appendix B is included as an example to illustrate the input and output we selected for our simulation of the tick population doubling each generation. The first part summarizes the input data. The second part lists our desired output—week number, number of adult female ticks, number of infested hosts, sterile-to-fertile ratio, and percentage of sterility—at weekly intervals. Then the number of adult female ticks is plotted against weeks in appendix figure B–1, and the same information is graphed logarithmically in appendix figure B–2. We concluded that our programing was accurate since our tick population doubled each generation. Simulations of hybrid release and of insecticide treatments were similarly successful.

Using this simulation model, we first attempted to duplicate the theoretical density reductions resulting from the release of sterile hybrids as described by Osburn and Knipling (1982). We will not go into a detailed description of the theory and potential of this genetic mechanism

Table 1.—Results of sterile-hybrid release in two simulations by 3-month period[1]

3-month period	Number of infested host animals in simulation—		Sterile-to-fertile ratio in simulation—		Percentage of sterility in simulation—	
	1	2	1	2	1	2
1	709	500	2.9 : 1	5.0 : 1	61	83
2	274	167	6.3 : 1	5.7 : 1	86	85
3	42	49	5.6 : 1	5.2 : 1	83	84
4	8	16	6.5 : 1	5.1 : 1	82	84
5	2	5	4.9 : 1	5.0 : 1	75	83
6	1	2	5.0 : 1	5.0 : 1	80	83
7	<1	<1	....	....	....	....

[1]Simulation 1 is the one done for this study; simulation 2 was done by Osburn and Knipling (1982).

Table 2.—The effect of interval of release on sterility and on the number of infested hosts by 3-month period

3-month period	Release interval (weeks)					
	1		2		4	
	No. of infested hosts	Percentage of sterility	No. of infested hosts	Percentage of sterility	No. of infested hosts	Percentage of sterility
1	709	61	704	62	710	60
2	274	86	275	86	322	83
3	42	83	43	81	74	82
4	8	82	9	79	19	74
5	2	75	3	75	8	73
6	1	80	1	72	5	67

for control. We repeat only that male hybrids resulting from crosses between *B. microplus* and *B. annulatus* are sterile and that hybrid females are fertile. The hybrid females that backcross to males of either species produce the same type of sterile males and fertile females for four to seven generations. Data from the simulation designed to mimic the results of Osburn and Knipling (table 1) show a close correspondence to these results in the decrease in number of infested hosts and in the sterile-to-fertile ratio and the percentage of sterility. The only differences appear in the first two 3-month periods and are easily understood. In both models, the population is doubling each generation. But in our simulation, hybrid ticks are released weekly and require some time to build up to an equilibrium density. So at first, both the wild tick population and the number of infested host animals increase under the influence of the doubling growth rate before the genetic-control mechanism brings about sufficient sterility to cause population reduction. These results confirm our belief in the theoretical accuracy of this model.

To evaluate the effect of increasing the interval of time between releases of the hybrids, we ran two additional simulations in which the number of hybrids released was the same, but the releases were made at 2- or 4-week intervals rather than every week. Table 2 shows the decrease in infested host animals and the average sterility for release intervals of 1, 2, and 4 weeks. Differences between results for 1- and 2-week release intervals were slight. But increasing the interval to 4

weeks decreased somewhat the effectiveness of the control strategy.

Our final simulations were conducted on the effectiveness of insecticide treatments to eliminate low-level tick populations. The method of insecticide treatment that was simulated was animal dipping or spraying where only the parasitic stages on the host would be subject to the treatment. We assumed the same level of effectiveness of the treatment (in percentage of kill) on the larvae, nymphs, and adults. In practice, larvae and nymphs are more susceptible than adults to insecticides. But using the same level of kill for all stages would make the outcome of the simulation a much more conservative estimate of the real situation. These simulations are not intended to indicate what would occur under real conditions. Obviously, it is much easier to control a tick population in a computer where all individuals are readily accessible in the storage registers of the computer than in actual tick populations where infested animals or wild hosts may be missed. But the use of such simulations by those familiar with tick control may provide insights into how the variables affect the outcome of the treatments.

We conducted several simulations in which we varied the percentage of kill of larvae, nymphs, and adults on the host by setting it at 99.9, 99, 95, 90, and 70 (table 3). In the first series at these levels of kill, 18 treatments applied at 2-week intervals were programed over a 9-month period. Then 12 treatments at 3-week intervals over the same period were simulated. Finally, a series of treatments at 2-week intervals for 5 months (10 treatments) was run. Table 3 presents a summary of these simulations expressed as the number of weeks to theoretical elimination of the tick populations. We reiterate that these are theoretical calculations and do not apply to practical problems of tick control or elimination under natural conditions. But it is interesting to note that, when 95% kill of ticks on all animals was obtained, elimination occurred in 7 weeks with the 2-week interval and in 9 weeks with the 3-week interval. When more data are available on the values of key parameters in determining the dynamics of tick populations, such simulations would be useful in designing treatment regimes for control or elimination and in predicting actual outcome of the treatments. In the meantime, the model and simulations should be useful in highlighting important key parameters. Verification of the output of the model with field trials would lend support to its usefulness in planning and prediction.

Table 3.—Time to theoretical population elimination with insecticide treatments when level of effectiveness, interval of treatment, and number of treatments were varied

Number of treatments	Treatment interval (weeks)	Number of weeks to theoretical 0 population with percentage of kill at—				
		99.9	99	95	90	70
18	2	5	5	7	9	(1)
12	3	5	6	9	(1)	(1)
10	2	5	5	(1)	(1)	(1)

[1]Total elimination did not occur.

REFERENCES

Graham, O. H., and Hourrigan, J. L.
1977. Eradication programs for the arthropod parasites of livestock. J. Med. Entomol. 13: 629-658.

Haile, D. G., and Weidhaas, D. E.
1977. Computer simulation of mosquito populations (*Anopheles albimanus*) for comparing the effectiveness of control technologies. J. Med. Entomol. 13: 553-567.

Hitchcock, L. F.
1955. Studies on the parasitic stages of the cattle tick, *Boophilus microplus* (Canestrini) (Acarina : Ixodidae). Aust. J. Zool. 3: 145-155.

Osburn, R. L., and Knipling, E. F.
1982. The potential use of sterile hybrid *Boophilus* ticks (Acari : Ixodidae) as a supplemental eradication technique. J. Med. Entomol. 19: 637-644.

Sutherst, R. W., and Dallwitz, M. J.
1979. Progress in the development of a population model for the cattle tick *Boophilus microplus*. Proc. Int. Congr. Acarol. (1974), 4th, pp. 557-563.

Sutherst, R. W., and Wharton, R. H.
1973. Preliminary considerations of a population model for *Boophilus microplus* in Australia. Proc. Int. Congr. Acarol. (1971), 3d, pp. 797-801.

Sutherst, R. W.; Wharton, R. H.; and Utech, K. B. W.
1978. Guide to studies on tick ecology. Aust. Commonw. Sci. Ind. Res. Organ. Div. Entomol. Tech. Bull. 14, 59 pp.

Thompson, G. D.; Osburn, R. L.; Davey, R. B.; and Price, M. A.
1981. The dynamics of hybrid sterility between *Boophilus annulatus* and *B. microplus* (Acari : Ixodidae) through successive generations. J. Med. Entomol. 18: 413-418.

APPENDIX A.—FORTRAN IV COMPUTER PROGRAM FOR *BOOPHILUS* TICK SIMULATIONS

```
//TICK JOB (7000,1102,60,20),'ARISTEA GANZOS',CLASS=5
/*PASSWORD
/*JOBPARM HEADER=USDA
//  EXEC WATFIV
$JOB
C
C
C  ***************************************************************
C  *                                                             *
C  *               TITLE:   TICK SIMULATION MODEL                *
C  *                                                             *
C  ***************************************************************
C
C
      DIMENSION RRLAR(13),XLAR(13),TIK(3000),TICK(3000),XX(80),
     $SCALE(80),FEMAL(80),SLAR(13)
C
C
C  ****************************************************************
C  *                                                              *
C  *  RRLAR,XLAR AND SLAR PROVIDE FOR RELEASED STERILE HYBRIDS,   *
C  *  WILD TICKS AND PROGENY OF STERILE HYBRIDS RESPECTIVELY.     *
C  *                                                              *
C  *  HOST= THE NUMBER OF ANIMALS;  TOTIK= THE NUMBER OF ADULT    *
C  *  MALE AND FEMALE TICKS;  INFHOS= THE NUMBER OF ANIMALS       *
C  *  INFESTED WITH TICKS;  AVTIK1 AND AVTIK2= AVERAGE NUMBER     *
C  *  OF TICKS PER INFESTED OR TOTAL ANIMALS; R,RR AND RRR=       *
C  *  RATIO STERILE HYBRIDS TO FERTILE MALES, FERTILITY AND       *
C  *  STERILITY RESPECTIVELY;  EGGSF= NUMBER OF MALE AND FEMALE   *
C  *  EGGS PER OVIPOSITION; VALUES WITH SUV ARE SURVIVAL FOR      *
C  *  INDICATED STAGES (EG= EGGS, LAI= LARVAE ON GROUND, LNA=     *
C  *  LARVAE, NYMPHS AND ADULTS ON HOST, AND ADF FOR ADULT        *
C  *  FEMALES ON GROUND);  WITH THESE SURVIVALS AND PICUP         *
C  *  (FRACTIONAL PICUP OF LARVAE BY HOSTS) GROWTH RATE IS 2X     *
C  *  XMAXLA IS THE MAXIMUM NUMBER OF WEEKS THE LARVAE SURVIVE    *
C  *  ON THE GROUND.                                              *
C  *                                                              *
C  ****************************************************************
C
C
      HOST=1000
      TOTIK=1000
      GG=0.5
      INFHOS=HOST*GG
      NONINF=HOST-INFHOS
      AVTIK1=TOTIK/HOST
      AVTIK2=TOTIK/INFHOS
      R1=1
      R2=1
      R3=1
      RR1=1
      RR2=1
      RR3=1
      RRR1=1
      RRR2=1
```

```
      RRR3=1
      XLAM=1.0933
      EGGSF=3000
      SUVEG=0.8660
      SUVLAI=0.495742
      SUVLNA=0.6299605
      SUVADF=0.5625
      XMAXLA=12
      PICUP=0.0114736
      N=2
      SPICUP=PICUP*N
C
C   ****************************************************************
C   *                                                              *
C   *  X IS USED TO REQUEST INITIALIZATION WITH THE IF STATEMENT   *
C   *  NWKS= THE TOTAL NUMBER OF WEEKS TO RUN THE MODEL            *
C   *  INT=  THE INTERVAL OF INSECTICIDE TREATMENTS                *
C   *  TRTWK=WEEK TO START THE INSECTICIDE TREATMENT               *
C   *  LSTRT=LAST INSECTICIDE TREATMENT                            *
C   *  ZZZ=  1- MORTALITY DUE TO INSECTICIDE TREATMENT             *
C   *  RLWK= THE WEEK TO START RELEASE                             *
C   *  RLINT=THE INTERVAL FOR RELEASES IN WEEKS                    *
C   *                                                              *
C   *  FOR NO INSECTICIDE TREATMENT SET TRTWK=0                    *
C   *  FOR NO HYBRID RELEASE SET RLWK=0                            *
C   *                                                              *
C   ****************************************************************
C
C
      NWKS=78
      INT=3
      X=1
      TRTWK=0
      LSTRT=3
      ZZZ=.01
      YYY= (1-ZZZ) * 100
      RLWK=0
      RLINT=1
      LSRL=78
      IF (X) 20,20,5
C
C   ***************************************************************
C   *     INITIALIZE THE WILD TICK DIMENSION- NEXT 14 STATEMENTS  *
C   ***************************************************************
C
C
 5    ADULF=TOTIK/2
      ADULM=ADULF
      ADULT1=ADULF * SUVLNA/XLAM
      ADULT2=ADULT1 * SUVADF/XLAM
      ADULT3=ADULT2 * SUVADF/XLAM
      EGG1=ADULT3 * EGGSF
      EGG2= EGG1 * SUVEG/XLAM
      XLAR(1)= EGG2 * SUVEG/XLAM
      SUM=0
```

```
           DO 10 I=1,12
              XLAR(I+1)=XLAR(I) * SUVLAI/XLAM
 10           SUM=SUM+XLAR(I)
              XLARHI= SUM * PICUP
              XNYMPH= XLARHI * SUVLNA/XLAM
            REL=XLAR(1)* 2.5
            SEGG1=0
            SEGG2=0
            DO 15 J=1,13
               RRLAR(J)=0
 15            SLAR(J)=0
               SLARH=0
               SNYMPH=0
               SADULF=0
               SADULM=0
               SADUL1=0
               SADUL2=0
               SADUL3=0
C
               DO 16 J=1,NWKS
                  XX(J)=J
                  SCALE(J)=0.0
 16               FEMAL(J)=0.0
C
C
C  **************************************************************
C  *                                                            *
C  *  WRITE OUT THE INITIAL CONDITIONS                          *
C  *  START THE WEEKLY ADJUSTMENTS FOR ALL THE TICKS IN THE     *
C  *  POPULATION.  FIRST MOVE THE WILD TICKS THROUGH THE WEEK   *
C  *  PERIOD.                                                   *
C  *                                                            *
C  **************************************************************
C
C
            PRINT 700
 700        FORMAT('1','INITIAL CONDITIONS-- WEEK 0')
            PRINT 701,HOST
 701        FORMAT('0',5X,'NUMBER HOST=  ',F6.1)
            PRINT 702, TOTIK
 702        FORMAT('0',5X,'TOTAL TICKS -MALE & FEMALE = ',F6.1)
            PRINT 703, GG
 703        FORMAT('0',5X,'PROPORTION OF HOSTS INFECTED = ',F4.2)
           PRINT 704, EGGSF
 704       FORMAT('0',5X,'FECUNDITY--TOTAL EGGS/FEMALE = ',F6.1)
           PRINT 705
 705       FORMAT('-','WEEKLY SURVIVAL RATES : ')
           PRINT 706
 706       FORMAT('0',5X,'EGGS',7X,'LARVAE OFF HOST',5X,'ALL FO'
          $,'RMS ON HOSTS',5X,'ADULT FEMALES OFF HOST')
           PRINT 707, SUVEG,SUVLAI,SUVLNA,SUVADF
 707       FORMAT('0',5X,F6.4,8X,F8.6,17X,F9.7,13X,F6.4)
           PRINT 708,PICUP
 708       FORMAT('-',5X,'LARVAL PICKUP BY HOSTS(FRACTION) = ',F10.7)
           PRINT 709
```

```
709  FORMAT('-','INSECTICIDE TREATMENTS')
     PRINT 710
710  FORMAT('0',5X,'INITIAL TREATMENT WEEK',5X,'TREATME'
    $,'NT INTERVAL')
     PRINT 610,TRTWK,INT
610  FORMAT('0',15X,F3.1,21X,I1)
     PRINT 711
711  FORMAT('-',5X,'FINAL TREATMENT WEEK',5X,'TREATME'
    $,'NT MORTALITY ALL FORMS ON HOSTS')
     PRINT 611,LSTRT,YYY
611  FORMAT('0',15X,I2,30X,F5.1)
     PRINT 712
712  FORMAT('-','HYBRID RELEASE')
     PRINT 713
713  FORMAT('0',5X,'INITIAL RELEASE WEEK',5X,'RELEASE INTER'
    $,'VAL',5X,'FINAL RELEASE WEEK',5X,'NUMBER REALESED')
     PRINT 714,RLWK,RLINT,LSRL,REL
714  FORMAT('0',15X,F3.1,18X,F3.1,20X,I3,14X,F10.2)
     PRINT 715
715  FORMAT('-','INITIAL WILD TICK LIFE TABLE')
     PRINT 716,EGG1,EGG2
716  FORMAT('0',F14.1,1X,F14.1)
     PRINT 717,(XLAR(I),I=1,7)
717  FORMAT('0',F14.1,1X,F14.1,1X,F14.1,1X,F14.1,1X,F14.1,
    $1X,F9.1,1X,F9.1)
     PRINT 617,(XLAR(I),I=8,13)
617  FORMAT(' ',F9.1,1X,F9.1,1X,F9.1,1X,F9.1,1X,F9.1,
    $1X,F9.1)
     PRINT 718,XLARHI
718  FORMAT('0',25X,F8.1)
     PRINT 719,XNYMPH
719  FORMAT('0',25X,F8.1)
     PRINT 720,ADULF
720  FORMAT('0',25X,F8.1)
     PRINT 721,ADULM
721  FORMAT('0',25X,F8.1)
     PRINT 722,ADULT1,ADULT2,ADULT3
722  FORMAT('0',25X,F8.1,9X,F8.1,9X,F8.1)
     PRINT 800
800  FORMAT('1','WEEK',3X,'ADULT  FEMALES',3X,'# INFEST'
    $,'ED HOSTS',3X,'STERILE/FERTILE',3X,'% STERILITY')
20     DO 80 K=1,NWKS
           ADULT3= ADULT2 * SUVADF
           ADULT2= ADULT1 * SUVADF
           ADULT1= ADULF *SUVLNA
           ADULF= XNYMPH * (SUVLNA/2)
           ADULM= XNYMPH * (SUVLNA/2)
           XNYMPH= XLARHI * SUVLNA
           SUM=0
       DO 30 I=1,11
           II=12-I+1
           XLAR(II)= XLAR(II-1) * SUVLAI
30     SUM=SUM + XLAR(II) * PICUP
       XLAR(1)= EGG2 * SUVEG
       XLARHI= SUM+ XLAR(1) * PICUP
```

```
      CHECK=XLARHI
      EGG2= EGG1 * SUVEG
      EGG1= ADULT3 * EGGSF
C
C
C    ***************************************************************
C    *                                                             *
C    *  DELAY THE STERILITY EFFECT FOR THREE WEEKS TO APPLY        *
C    *  TO THE EGGS RATHER THAN THE FEMALE ADULTS.                 *
C    *                                                             *
C    ***************************************************************
C
C
      R3=R2
      R2=R1
      RR3=RR2
      RR2=RR1
      RRR3=RRR2
      RRR2=RRR1
C
C
C    ***************************************************************
C    *   MOVE HYBRID PROGENY AHEAD ONE WEEK                        *
C    ***************************************************************
C
C
      SADUL3= SADUL2 * SUVADF
      SADUL2 = SADUL1 * SUVADF
      SADUL1= SADULF * SUVLNA
      SADULF= SNYMPH * (SUVLNA/2)
      SADULM= SNYMPH * (SUVLNA/2)
      SNYMPH= SLARH* SUVLNA
      SUM=0
C
      DO 40 J=1,11
         JJ=12-J+1
         SLAR(JJ)=SLAR(JJ-1) * SUVLAI
 40   SUM= SUM + SLAR(JJ) * PICUP
      SLAR(1)= SEGG2 * SUVEG
C
C
C    **************************************************************
C    *  MOVE RELEASED HYBRID LARVAE AHEAD ONE WEEK                *
C    **************************************************************
C
C
      DUM=0
      DO 50 N=1,11
            NN=12-N+1
            RRLAR(NN)=RRLAR(NN-1)*SUVLAI
            DUM=DUM+RRLAR(NN)*SPICUP
 50   CONTINUE
      RRLAR(1)=0
```

```
C
C    ***************************************************************
C    *  USED TO SPECIFY THE MAXIMUM NUMBER OF RELEASES              *
C    ***************************************************************
C
           IF (K .GT. LSRL) RLWK=Ø
C
C    ***************************************************************
C    *  USED TO SPECIFY WHEN TO RELEASE                             *
C    ***************************************************************
C
           IF (K .EQ. RLWK) THEN DO
             RRLAR(1)=REL
             RLWK=RLWK + RLINT
           ENDIF
           SLARH=SUM+SLAR(1) * PICUP
           DISTIK=SLARH
           SLARH=SLARH+DUM+RRLAR(1)*SPICUP
C
C    ***************************************************************
C    *  PROVIDING FOR WASTING RELEASED AND PROGENY HYBRIDS          *
C    ***************************************************************
C
           SLARH=(SLARH/HOST)*INFHOS
           SEGG2= SEGG1 * SUVEG
           SEGG1= SADUL3 * EGGSF
           IF (ADULM .EQ. Ø.Ø) R1=9999999
           IF (ADULM .GT. Ø.Ø) R1=SADULM/ADULM
           IF ((ADULM .EQ. Ø.Ø) .AND. (SADULM .EQ. Ø.Ø)) THEN DO
              RR1=9999999
              RRR1=9999999
           ELSE DO
              RR1= ADULM/(SADULM+ ADULM)
              RRR1= SADULM/(SADULM+ADULM) * 1ØØ
           ENDIF
           EGG1= EGG1 * RR3
           SEGG1= SEGG1 * RR3
C
C    ***************************************************************
C    *  CALLING FOR THE INSECTICIDE TREATMENT AND SPECIFYING        *
C    *  TIMING AND NUMBER OF TREATMENTS                             *
C    ***************************************************************
C
          IF(K .GT. LSTRT) TRTWK=Ø
          IF (TRTWK .EQ. K) CALL XINS(XLARHI,ZZZ,XNYMPH,ADULF,
         $ADULM,SLARH,SNYMPH,SADULF,SADULM,TRTWK,INT)
          GG= ADULF/HOST
          INFHOS=GG * HOST
          IF (GG .EQ. 1) INFHOS=HOST
          IF (GG .GT. 1) INFHOS=HOST
          SCALAD=ADULF
          IF (SCALAD .LT. 1) SCALAD=1
          SCALE(K)=ALOG1Ø(SCALAD)
          FEMAL(K)=ADULF
```

```
      PRINT 801,K,ADULF,INFHOS,R1,RRR1
801   FORMAT(' ',1X,I2,7X,F9.2,10X,I4,10X,F8.2,10X,F8.2)
80    IF (CHECK .LT. 2)  GO TO 81
81    XNWKS=NWKS
      CALL PLOT1(1,26,5,61,12)
      CALL PLOT2(TICK,80.0,0.0,5000.0,0.0)
      CALL PLOT3(1HT,XX(1),FEMAL(1),NWKS)
      WRITE(6,888)
888   FORMAT(1H1,30X,'TICK MODEL')
      WRITE(6,889)
889   FORMAT(20X,'TOTAL NATIVE ADULT FEMALES VS. TIME(WEEKS)')
      CALL PLOT4(20,20HNUMBER ADULT FEMALES)
      WRITE(6,665)
665   FORMAT('0',30X,'TIME(WEEKS)')
      CALL PLOT1(1,26,5,61,12)
      CALL PLOT2(TIK,80.0,0.0,5.0,0.0)
      CALL PLOT3(1HT,XX(1),SCALE(1),NWKS)
      WRITE(6,666)
666   FORMAT(1H1,30X,'TICK MODEL')
      WRITE(6,667)
667   FORMAT(20X,'LOG 10(TOTAL NATIVE ADULT FEMALES)')
      WRITE(6,650)
650   FORMAT(30X,' VS. TIME(WEEKS)')
      CALL PLOT4(24,24HLOG NUMBER ADULT FEMALES)
      WRITE(6,668)
668   FORMAT('0',30X,'TIME(WEEKS)')
      WRITE(6,669)
669   FORMAT('1')
     STOP
     END
     SUBROUTINE XINS(XLARHI,ZZZ,XNYMPH,ADULF,ADULM,SLARH,SNYMPH,
    $SADULF,SADULM,TRTWK,INT)
C
      XLARHI= XLARHI * ZZZ
      XNYMPH= XNYMPH * ZZZ
      ADULF= ADULF * ZZZ
      ADULM= ADULM * ZZZ
      SLARH= SLARH * ZZZ
      SNYMPH= SNYMPH * ZZZ
      SADULF= SADULF * ZZZ
      SADULM= SADULM * ZZZ
      TRTWK= TRTWK + INT
      RETURN
      END
$ENTRY
/*EOJ
END OF WORK FILE
```

APPENDIX B.—SIMULATION OF A TICK POPULATION WITHOUT CONTROL TREATMENTS

INPUT

INITIAL CONDITION -- WEEK = Ø

NUMBER OF HOSTS = 1ØØØ.Ø

TOTAL TICKS (MALE & FEMALE) = 1ØØØ.Ø

PROPORTION OF HOSTS INFECTED = Ø.5Ø

FECUNDITY (TOTAL EGGS / FEMALE) = 3ØØØ.Ø

WEEKLY SURVIVAL RATES:

EGGS	LARVAE OFF HOST	ALL FORMS ON HOSTS	ADULT FEMALES OFF HOST
Ø.866Ø	Ø.495742	Ø.62996Ø5	Ø.5625

LARVAL PICKUP BY HOSTS(FRACTION) = Ø.Ø114736

INSECTICIDE TREATMENTS

INITIAL TREATMENT WEEK	TREATMENT INTERVAL
Ø.Ø	--

FINAL TREATMENT WEEK	TREATMENT MORTALITY ALL FORMS ON HOSTS
--	--

HYBRID RELEASE

INITIAL RELEASE WK	RELEASE INTERVAL	FINAL RELEASE WK	# RELEASED
Ø.Ø	--	--	--

INITIAL WILD TICK LIFE TABLE:

EGGS--	228786	181221					
LARVAE OFF HOSTS--	143545	65Ø88	29513	13382	6Ø68	2751	1247
	565	256	116	52	23	1Ø	
LARVAE ON HOSTS--	3Ø13						
NYMPHS ON HOSTS--	736						
ADULT FEMALES ON HOSTS--	5ØØ						
ADULT MALES ON HOSTS--	5ØØ						
ADULT FEMALES OFF HOSTS--	288	148	76				

OUTPUT

WEEK	ADULT FEMALES	# INFESTED HOSTS	STERILE/FERTILE	% STERILITY
1	546.85	546	0.00	0.00
2	597.87	597	0.00	0.00
3	653.66	653	0.00	0.00
4	714.64	714	0.00	0.00
5	781.32	781	0.00	0.00
6	854.22	854	0.00	0.00
7	933.92	933	0.00	0.00
8	1021.26	1000	0.00	0.00
9	1116.64	1000	0.00	0.00
10	1220.88	1000	0.00	0.00
11	1334.81	1000	0.00	0.00
12	1459.36	1000	0.00	0.00
13	1595.53	1000	0.00	0.00
14	1744.39	1000	0.00	0.00
15	1907.36	1000	0.00	0.00
16	2085.53	1000	0.00	0.00
17	2280.26	1000	0.00	0.00
18	2493.11	1000	0.00	0.00
19	2725.79	1000	0.00	0.00
20	2980.14	1000	0.00	0.00
21	3258.21	1000	0.00	0.00
22	3562.43	1000	0.00	0.00
23	3895.13	1000	0.00	0.00
24	4258.87	1000	0.00	0.00
25	4656.49	1000	0.00	0.00
26	5091.14	1000	0.00	0.00
27	5566.28	1000	0.00	0.00
28	6085.71	1000	0.00	0.00
29	6653.79	1000	0.00	0.00
30	7275.05	1000	0.00	0.00
31	7954.38	1000	0.00	0.00
32	8697.08	1000	0.00	0.00
33	9508.99	1000	0.00	0.00
34	10396.56	1000	0.00	0.00
35	11366.84	1000	0.00	0.00
36	12427.79	1000	0.00	0.00
37	13588.00	1000	0.00	0.00
38	14856.68	1000	0.00	0.00
39	16243.82	1000	0.00	0.00
40	17760.36	1000	0.00	0.00
41	19418.29	1000	0.00	0.00
42	21230.73	1000	0.00	0.00
43	23212.35	1000	0.00	0.00
44	25379.16	1000	0.00	0.00
45	27748.52	1000	0.00	0.00
46	30339.23	1000	0.00	0.00
47	33171.77	1000	0.00	0.00
48	36268.55	1000	0.00	0.00
49	39654.04	1000	0.00	0.00
50	43355.39	1000	0.00	0.00
51	47402.38	1000	0.00	0.00

52	51827.49	1000	0.00	0.00
53	56666.02	1000	0.00	0.00
54	61956.37	1000	0.00	0.00
55	67740.44	1000	0.00	0.00
56	74064.13	1000	0.00	0.00
57	80977.63	1000	0.00	0.00
58	88536.56	1000	0.00	0.00
59	96801.31	1000	0.00	0.00
60	105838.10	1000	0.00	0.00
61	115718.90	1000	0.00	0.00
62	126522.10	1000	0.00	0.00
63	138333.50	1000	0.00	0.00
64	151246.80	1000	0.00	0.00
65	165365.30	1000	0.00	0.00
66	180801.80	1000	0.00	0.00
67	197680.00	1000	0.00	0.00
68	216134.30	1000	0.00	0.00
69	236311.70	1000	0.00	0.00
70	258372.60	1000	0.00	0.00
71	282492.30	1000	0.00	0.00
72	308862.80	1000	0.00	0.00
73	337695.00	1000	0.00	0.00
74	369218.80	1000	0.00	0.00
75	403686.10	1000	0.00	0.00
76	441372.00	1000	0.00	0.00
77	482576.40	1000	0.00	0.00
78	527626.40	1000	0.00	0.00


TICK MODEL
TOTAL NATIVE ADULT FEMALES VS. TIME(WEEKS)
NUMBER ADULT FEMALES
5000.000
4000.000
3000.000
2000.000
1000.000
0.000
0.00
16.00
32.00
48.00
64.00
80.00
TIME(WEEKS)

FIGURE B-1

U. S. DEPARTMENT OF AGRICULTURE
AGRICULTURAL RESEARCH SERVICE
SOUTHERN REGION
P. O. BOX 53326
NEW ORLEANS, LOUISIANA 70153

OFFICIAL BUSINESS
PENALTY FOR PRIVATE USE, $300

POSTAGE AND FEES PAID
U. S. DEPARTMENT OF
AGRICULTURE
AGR 101

U.S.MAIL

```
                            TICK MODEL
                LOG 10(TOTAL NATIVE ADULT FEMALES)
                         VS. TIME(WEEKS)
       5.000 +-----------+-----------+-----------+------TT---+-----------+
             |           |           |           | TTTT      |           |
  L          |           |           |          TTTT         |           |
  O          |           |           |       TTTT|           |           |
  G          |           |           |   TTTT    |           |           |
       4.000 +-----------+----------TTTTT--------+-----------+-----------+
  N          |           |       TTTT|           |           |           |
  U          |           |   TTTT    |           |           |           |
  M          |           TTTT        |           |           |           |
  B          |       TTTT|           |           |           |           |
  E    3.000 +----TTTT---+-----------+-----------+-----------+-----------+
  R          |TTTT       |           |           |           |           |
             |           |           |           |           |           |
  A          |           |           |           |           |           |
  D          |           |           |           |           |           |
  U    2.000 +-----------+-----------+-----------+-----------+-----------+
  L          |           |           |           |           |           |
  T          |           |           |           |           |           |
             |           |           |           |           |           |
  F          |           |           |           |           |           |
  E    1.000 +-----------+-----------+-----------+-----------+-----------+
  M          |           |           |           |           |           |
  A          |           |           |           |           |           |
  L          |           |           |           |           |           |
  E          |           |           |           |           |           |
  S    0.000 +-----------+-----------+-----------+-----------+-----------+
            0.00       16.00       32.00       48.00       64.00       80.00

                            TIME(WEEKS)
```

FIGURE B-2

☆ U.S. GOVERNMENT PRINTING OFFICE: 1983-769-037:7